WORKFILE DMR89 U 5
MATRIX(5,5) MSAM

FOR %AC ACT COM HHD ROW TOT

VECTOR(5) V{%AC}

NEXT

VACT(3)=100

VACT(5)=@SUM(VACT)

VCOM(1)=75

VCOM(4)=25

VCOM(5)=@SUM(VCOM)

VHHD(2)=100

VHHD(5)=@SUM(VHHD)

VROW(1)=25

VROW(5)=@SUM(VROW)

!COL=1

FOR %AC ACT COM HHD ROW

COLPLACE(MSAM,V{%AC},!COL)

DELETE V{%AC}

!COL=!COL+1

NEXT

!NRWS=@ROWS(MSAM)

FOR !R=1 TO (!NRWS-1)

ROWVECTOR RV{!R}=@ROWEXTRACT(MSAM, !R)

VTOT(!R)=@SUM(RV{!R})

DELETE RV{!R}

NEXT

COLPLACE(MSAM,VTOT, !NRWS)

DELETE VTOT

MODEL DMR

DMR.APPEND PX=(PE*XE + PD*XD)/XS 

DMR.APPEND PD=(PQ*QQ - PM*QM)/XD 

DMR.APPEND PE=EXR*PWE 

DMR.APPEND XE = XD * ( (PE/PD) * (1 - alpha) / alpha )^(1 / phi) 

DMR.APPEND XD=XS-XE 

DMR.APPEND PM=EXR*PWM 
DMR.APPEND QM = XD * ( (PD / PM)*beta / (1 - beta) )^(1 / (1 + rho)) 

DMR.APPEND QQ=bq*( beta*QM^(-rho) + (1-beta)*XD^(-rho) )^(-1/rho) 

DMR.APPEND YH=PX*XS + EXR*BOT 

DMR.APPEND EXR*BOT=(PM*QM -PE*XE )

DMR.APPEND RPD=EXR/PD

DMR.APPEND RPX=EXR/PX

FOR %VR BOT EXR PD PE PM PQ PWE PWM PX QQ QM RPD RPX XD XE XS YH omega sigma

SERIES {%VR}
NEXT

BOT=0

PD=1

PWE=1

PWM=1

XD=MSAM(1,2)

EXR=1

PE=1

PM=1

PQ=1

PX=1

RPD=EXR/PD

RPX=EXR/PX

QQ=MSAM(2,5)

QM=MSAM(4,2)
XD=MSAM(1,2)

XE=MSAM(1,4)

XS=MSAM(1,5)

YH=MSAM(3,5)

MODEL CALIBER

omega.fill 0.2, 0.5, 2, 5, 5000000 
sigma.fill 0.2, 0.5, 2, 5, 5

CALIBER.APPEND rho=(1/sigma) - 1

CALIBER.APPEND phi=(1/omega) +1

CALIBER.APPEND alpha=1/(1 + (PD/PE)*(XE/XD)^(phi ))

CALIBER.APPEND ax = XS/(alpha*XE^phi + (1-alpha)*XD^phi )^(1/phi) 

CALIBER.APPEND beta=((PM/PD)*(QM/XD)^(1+rho) )/(1 + (PM/PD)*(QM/XD)^(rho+1)) 

CALIBER.APPEND bq = QQ/(beta*QM^(-rho) + (1-beta)*XD^(-rho) )^(-1/rho) 

CALIBER.SCENARIO ACTUALS

CALIBER.SOLVE(s=d, d=s, o=n)

DMR.SOLVEOPT(s=d, d=s,c=1e-15, o=n)

DMR.SCENARIO(c) BASELINE 

SOLVE DMR

DMR.MAKEGROUP(a) BASEGRP @ENDOG

FREEZE(BASELINE) BASEGRP

